# Using Mathematica to find series expansions for partial derivatives of the generalized Riemann zeta function

I am trying to use Mathematica to find a suitable series expansion for the expression $$\zeta ^{(1,0)}\left(-1,1-\frac{i}{2}\right) - \zeta^{(1,0)}\left(-1,1+\frac{i}{2}\right),$$ which Mathematica evaluates numerically as 0.484427... times the imaginary unit $$i$$, where the superscripts in the expression displayed above indicate that the generalized Riemann zeta function $$\zeta(s,a)$$ is being differentiated with respect to $$s$$, and then evaluated at the indicated arguments. It seems that some issues arise if we consider how this numerical evaluation is obtained through a direct application of the definition of the function $$\zeta(s,a)$$, which Mathematica defines so that $$\zeta(s, a) = \sum_{k=0}^{\infty} (k + a)^{-s},$$ omitting the cases whereby $$k + a = 0$$. It would seem that we should have that $$\zeta^{(1,0)}(t,b) = \sum _{k=0}^{\infty } -(b+k)^{-t} \log (b+k)$$ whenever the above series converges. However, the series $$\sum _{k=0}^{\infty } \left(\left(-1+\frac{i}{2}\right)-k\right) \log \left(\left(1-\frac{i}{2}\right)+k\right)$$ does not converge, but Mathematica evaluates $$\zeta ^{(1,0)}\left(-1,1-\frac{i}{2}\right)$$ numerically as $$(-0.224051...) + (0.242213...) i$$. So, since the above series does not converge, why does Mathematica evaluate $$\zeta ^{(1,0)}\left(-1,1-\frac{i}{2}\right)$$ as having a finite real part and a finite imaginary part? What does the expression $$\zeta ^{(1,0)}\left(-1,1-\frac{i}{2}\right)$$ even mean if the corresponding series does not converge? What would be a suitable way of writing this expression as an infinite sum derived from the series definition for the generalized Riemann zeta function?

Thank you.

• The definition of zeta as an infinite sum is only valid in the domain where the sum converges. Outside that domain, the function is continued analytically. This is similar to the sum of the geometric series which has a finite radius of convergence but has a simple rational function expression except at one point which is a simple pole. Feb 10 '19 at 0:23

As @somos says you need an alternative series expansion that converges for your point of interest. The one at http://functions.wolfram.com/ZetaFunctionsandPolylogarithms/Zeta2/06/05/01/01/0001/ is more useful in this case. Summing only for $$k\le n$$:

myzeta[s_, a_, n_] := Sum[(-1)^j*Binomial[k,j]*(a+j)^(1-s)/((k+1)*(s-1)),
{k,0,n}, {j,0,k}]


Taking the first derivative of each term with respect to $$s$$:

myzeta1[s_, a_, n_] := -Sum[(-1)^j*Binomial[k,j]*(a+j)^(1-s)*(Log[a+j]+1/(s-1))/((k+1)*(s-1)),
{k,0,n}, {j,0,k}]


Your formula is thus, for example with $$k\le10$$:

With[{n = 10}, myzeta1[-1, 1-I/2, n] - myzeta1[-1, 1+I/2, n]] // N
(* 0. + 0.490021 I *)


Replacing // N with // ComplexExpand gives you a formula that you can interpret as a series expression:

With[{n = 10}, myzeta1[-1, 1-I/2, n] - myzeta1[-1, 1+I/2, n]] // ComplexExpand
(* I (1/4 - 483/44 ArcTan[1/22] + 44289/440 ArcTan[1/20]
- (1631473 ArcTan[1/18])/3960 + (1045007 ArcTan[1/16])/1056
- (11394539 ArcTan[1/14])/7392 + (16353181 ArcTan[1/12])/10080
- (1303391 ArcTan[1/10])/1120 + (983509 ArcTan[1/8])/1760
- (541091 ArcTan[1/6])/3168 + (221209 ArcTan[1/4])/7392
- (83711 ArcTan[1/2])/36960 - (83711 Log[5/4])/55440
+ (221209 Log[17/4])/27720 - (541091 Log[37/4])/18480
+ (983509 Log[65/4])/13860 - (1303391 Log[101/4])/11088
+ (1257937 Log[145/4])/9240 - (876503 Log[197/4])/7920
+ 61471/990 Log[257/4] - 5051/220 Log[325/4]
+ 111/22 Log[401/4] - 1/2 Log[485/4]) *)